**Online appendix to accompany:**

Preacher, K. J., Zhang, Z., & Zyphur, M. J. (in press). Multilevel structural equation models for assessing moderation within and across levels of analysis. *Psychological Methods*.

In this appendix we provide Mplus syntax for testing a variety of multilevel moderation hypotheses. In what follows, L1 denotes Level-1 and L2 denotes Level-2. We use these codes to represent various multilevel moderation hypotheses, consistent with the manuscript:

A1: Within part of L1 moderator × Within part of L1 predictor
A2: Between part of L1 moderator × Within part of L1 predictor (cross-level interaction)
A3: Between part of L1 moderator × Between part of L1 predictor
B1: L2 moderator × Within part of L1 predictor (cross-level interaction)
B2: L2 moderator × Between part of L1 predictor
C: L2 moderator × L2 predictor
D: Between part of L1 moderator × L2 predictor

In models with an L1 predictor that has been decomposed into latent Within and Between components (i.e., by *not* declaring this variable on either the WITHIN= or the BETWEEN= lines), the %WITHIN% slope represents the within-cluster effect $\gamma_W$ and the %BETWEEN% slope represents the between-cluster effect $\gamma_B$.

In models with this same L1 predictor that has been declared as a WITHIN variable in Mplus (but its observed cluster mean is not included at the Between level), the %WITHIN% slope then becomes a composite of $\gamma_W$ and $\gamma_B$.

In models with this same L1 predictor declared as a WITHIN variable and the observed cluster mean is included in the Between part of the model, the %WITHIN% slope is the within-cluster effect $\gamma_W$ and the %BETWEEN% slope for the mean variable is interpreted as a contextual effect (the difference of the between- and within-cluster effects, or $\gamma_B - \gamma_W$).

In what follows, LMS = latent moderated structural equations, RCP = random coefficient prediction, and HS&B = High School and Beyond.

**Contents and associated files**

*Simulated examples*

1. Syntax for an A1 hypothesis using LMS
   [*sim.lms.a1.inp*, *sim.lms.a1.out*]
2. Syntax for an A2 hypothesis using LMS
   [*sim.lms.a2.inp*, *sim.lms.a2.out*]
3. Syntax for an A3 hypothesis using LMS
   [*sim.lms.a3.inp*, *sim.lms.a3.out*]

4. Syntax for A1 and A2 hypotheses using LMS
   [*sim.lms.a1a2.inp*, *sim.lms.a1a2.out*]
5. Syntax for a B1 hypothesis using LMS
   [*sim.lms.b1.inp*, *sim.lms.b1.out*]
6. Syntax for a B2 or D hypothesis using LMS
   [*sim.lms.b2d.inp*, *sim.lms.b2d.out*]
7. Syntax for an A1 hypothesis (random slope) using LMS
   [*sim.lms.a1.random.inp*, *sim.lms.a1.random.out*]
8. Syntax for a C hypothesis using LMS
   [*sim.lms.c.inp*, *sim.lms.c.out*]
9. Syntax for an A2 hypothesis using RCP
   [*sim.rcp.a2.inp*, *sim.rcp.a2.out*]
10. Syntax for two A2 hypotheses using RCP with a Bayes estimator
    [*sim.rcp.a2a2.bayes.inp*, *sim.rcp.a2a2.bayes.out*]
11. Syntax for an A3 hypothesis using RCP
    [*sim.rcp.a3.inp*, *sim.rcp.a3.out*]
12. Syntax for a B1 hypothesis using RCP
    [*sim.rcp.b1.inp*, *sim.rcp.b1.out*]
13. Syntax for a B2 or D hypothesis using RCP
    [*sim.rcp.b2d.inp*, *sim.rcp.b2d.out*]
14. Syntax for a C hypothesis using RCP
    [*sim.rcp.c.inp*, *sim.rcp.c.out*]

*Example (HS&B data)*

15. Syntax for B1 and B2 hypotheses using LMS
    [*hsab.lms.lms.b1b2.inp*, *hsab.lms.lms.b1b2.out*]
16. Syntax for a B1 hypothesis using RCP and a B2 hypothesis using LMS
    [*hsab.rcp.lms.b1b2.inp*, *hsab.rcp.lms.b1b2.out*]
17. Syntax for conflated model
    [*hsab.rcp.conflated.inp*, *hsab.rcp.conflated.out*]

*Extensions (HS&B data)*

18. Syntax for a 3-way interaction of one L1 and two L2 variables (RCP and product)
    [*hsab.rcp.3way.inp*, *hsab.rcp.3way.out*]
19. Syntax for a 3-way interaction of one L2 and two L1 variables (LMS only)
    [*hsab.lms.3way.inp*, *hsab.lms.3way.out*]
20. Syntax for an interaction of L1 and L2 components of the same variable
    [*hsab.rcp.automod.inp*, *hsab.rcp.automod.out*]

## Simulated examples

All syntax files in this section both generate and analyze data.

1. *Syntax for an A1 hypothesis using LMS*

```
TITLE: syntax for an A1 hypothesis using LMS, 1x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.a1.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
xzw | xw XWITH zw; xw WITH zw*.1;
yw ON xw*.1 zw*.3 xzw*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
xzw | xw XWITH zw; xw WITH zw*.1;
yw ON xw*.1 zw*.3 xzw*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
```

2. *Syntax for an A2 hypothesis using LMS*

```
TITLE: syntax for an A2 hypothesis using LMS, 1x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.a2.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
xz | xw XWITH zb; xw WITH zw*.1;
yw ON xw*.1 zw*.3 xz*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
xz | xw XWITH zb; xw WITH zw*.1;
yw ON xw*.1 zw*.3 xz*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
```

3. *Syntax for an A3 hypothesis using LMS*

```
TITLE: syntax for an A3 hypothesis using LMS, 1x(1–1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.a3.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.1 zw*.3; xw WITH zw*.1;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
xzb | xb XWITH zb; xb WITH zb*.1;
yb ON xb*.2 zb*.2 xzb*.2;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.1 zw*.3; xw WITH zw*.1;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
xzb | xb XWITH zb; xb WITH zb*.1;
yb ON xb*.2 zb*.2 xzb*.2;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
```

4. *Syntax for A1 and A2 hypotheses using LMS*

```
TITLE: Syntax for A1 and A2 hypotheses using LMS, 1x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.a1a2.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
xw WITH zw*.1;
xzw | xw XWITH zw;
xzb | xw XWITH zb;
yw ON xw*.1 zw*.3 xzw*.2 xzb*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 5;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
xw WITH zw*.1;
xzw | xw XWITH zw;
xzb | xw XWITH zb;
yw ON xw*.1 zw*.3 xzw*.2 xzb*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
```

5. *Syntax for a B1 hypothesis using LMS*

```
TITLE: syntax for a B1 hypothesis using LMS, 2x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
BETWEEN ARE z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.b1.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
s | yw ON xw;
xz | xw XWITH zb; yw ON xz*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1 s*0; zb WITH s*0;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.1]; s*.2;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
s | yw ON xw;
xz | xw XWITH zb; yw ON xz*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1 s*0; zb WITH s*0;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.1]; s*.2;
```

6. *Syntax for a B2 or D hypothesis using LMS*

```
TITLE: syntax for a B2 or D hypothesis using LMS, 2x(1-1) or 1x(2-1) design,
       note that although B2 and D are similar in analysis, they represent
       different research questions;
MONTECARLO:
NAMES ARE y x z;
BETWEEN ARE z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.b2d.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
xzb | xb XWITH zb; xb WITH zb*.1;
yb ON xb*.2 zb*.2 xzb*.2;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 3;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
xzb | xb XWITH zb; xb WITH zb*.1;
yb ON xb*.2 zb*.2 xzb*.2;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
```

7. *Syntax for an A1 hypothesis (random slope) using LMS*

```
TITLE: syntax for an A1 hypothesis (random slope) using LMS, 1x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.lms.a1.random.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
xzw | xw XWITH zw; xw WITH zw*.1;
y ON xw*.1 zw*.3; y*.7;
ywx BY; ywx ON xzw@1; ywx@0;
s | y ON ywx;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
y ON xb*.2 zb*.2; xb WITH zb*.1; y*.7;
[x@0 z@0 y*.1 xb*0 zb*0 s*.2]; s*.2;
s WITH y*0 xb*0 zb*0;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
xzw | xw XWITH zw; xw WITH zw*.1;
y ON xw*.1 zw*.3; y*.7;
ywx BY; ywx ON xzw@1; ywx@0;
s | y ON ywx;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
y ON xb*.2 zb*.2; xb WITH zb*.1; y*.7;
[x@0 z@0 y*.1 xb*0 zb*0 s*.2]; s*.2;
s WITH y*0 xb*0 zb*0;
```

8. *Syntax for a C hypothesis using LMS*

```
TITLE: syntax for a C hypothesis using LMS, 2x(2-1) design;
MONTECARLO:
NAMES ARE y x z;
BETWEEN ARE x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5723;
NREPS = 1;
SAVE IS sim.lms.c.dat;
MODEL POPULATION:
%WITHIN%
yw BY y@1; yw*.7; y@.01;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
xzb | xb XWITH zb; xb WITH zb*.1;
yb ON xb*.1 zb*.2 xzb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
yw BY y@1; yw*.7; y@.01;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
xzb | xb XWITH zb; xb WITH zb*.1;
yb ON xb*.1 zb*.2 xzb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1];
```

9. *Syntax for an A2 hypothesis using RCP*

```
TITLE: syntax for an A2 hypothesis using RCP, 1x(1–1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.rcp.a2.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
s | yw ON xw; xw WITH zw*.1;
yw ON zw*.3;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
s ON zb*.2; s@.01; s WITH yb@0 xb@0 zb@0;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 4;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
s | yw ON xw; xw WITH zw*.1;
yw ON zw*.3;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.1;
s ON zb*.2; s@.01; s WITH yb@0 xb@0 zb@0;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.1];
```

10. *Syntax for two A2 hypotheses using RCP with a Bayes estimator*

```
TITLE: syntax for two A2 hypotheses using RCP with Bayes, 1x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 20000;
NCSIZES = 1;
CSIZES = 800(25);
SEED = 5721;
NREPS = 1;
SAVE IS sim.rcp.a2a2.bayes.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
s1 | y ON xw; y*.7;
s2 | y ON zw; xw WITH zw*.1;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
xb WITH zb*.1; y*.3;
s1 ON zb*.2; s1@.01; y WITH s1@0;
s2 ON xb*-.2; s2@.01; y WITH s2@0;
y ON xb*.2 zb*.2; s1 WITH s2@0 xb@0 zb@0; s2 WITH xb@0 zb@0;
[x@0 z@0 xb*0 zb*0 y*.1 s1*.1 s2*.3];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS BAYES;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
s1 | y ON xw; y*.7;
s2 | y ON zw; xw WITH zw*.1;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
xb WITH zb*.1; y*.3;
s1 ON zb*.2; s1@.01; y WITH s1@0;
s2 ON xb*-.2; s2@.01; y WITH s2@0;
y ON xb*.2 zb*.2; s1 WITH s2@0 xb@0 zb@0; s2 WITH xb@0 zb@0;
[x@0 z@0 xb*0 zb*0 y*.1 s1*.1 s2*.3];
```

11. *Syntax for an A3 hypothesis using RCP*

```
TITLE: syntax for an A3 hypothesis using RCP, 1x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5723;
NREPS = 1;
SAVE IS sim.rcp.a3.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.2 zw*.3; xw WITH zw*.1;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
s | yb ON xb; s@.001;
s ON zb*.2; s WITH yb@0 xb@0 zb@0;
yb ON zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.2];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 5;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
zw BY z@1; zw*.7; z@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.2 zw*.3; xw WITH zw*.1;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
s | yb ON xb; s@.001;
s ON zb*.2; s WITH yb@0 xb@0 zb@0;
yb ON zb*.2; xb WITH zb*.1;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.2];
```

12. *Syntax for a B1 hypothesis using RCP*

```
TITLE: syntax for a B1 hypothesis using RCP, 2x(1-1) design;
MONTECARLO:
NAMES ARE y x z;
BETWEEN ARE z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 200(50);
SEED = 2;
NREPS = 1;
SAVE IS sim.rcp.b1.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
s | yw ON xw;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*1.5; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.0;
s ON zb*.2; s@.01; s WITH yb@0 xb@0 zb@0;
[x@0 y@0 xb*0 z@0 zb*0 yb*4 s*.6];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 6;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
s | yw ON xw;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*1.5; y@.01;
yb ON xb*.2 zb*.2; xb WITH zb*.0;
s ON zb*.2; s@.01; s WITH yb@0 xb@0 zb@0;
[x@0 y@0 xb*0 z@0 zb*0 yb*4 s*.6];
```

13. *Syntax for a B2 or D hypothesis using RCP*

```
TITLE: syntax for a B2 or D hypothesis using RCP, 2x(1-1) or 1x(2-1) design,
       note that although B2 and D are similar in analysis, they represent
       different research questions;
MONTECARLO:
NAMES ARE y x z;
BETWEEN ARE z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5721;
NREPS = 1;
SAVE IS sim.rcp.b2d.dat;
MODEL POPULATION:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON zb*.2; xb WITH zb*.1;
s | yb ON xb*.2; s@0;
s ON zb*.2; s WITH yb@0 xb@0;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 5;
MODEL:
%WITHIN%
xw BY x@1; xw*.7; x@.01;
yw BY y@1; yw*.7; y@.01;
yw ON xw*.2;
%BETWEEN%
xb BY x@1; xb*.7; x@.01;
zb BY z@1; zb*.7; z@.01;
yb BY y@1; yb*.7; y@.01;
yb ON zb*.2; xb WITH zb*.1;
s | yb ON xb*.2; s@0;
s ON zb*.2; s WITH yb@0 xb@0;
[x@0 z@0 y@0 xb*0 zb*0 yb*.1 s*.1];
```

14. *Syntax for a C hypothesis using RCP*

```
TITLE: syntax for a C hypothesis using RCP, 2x(2-1) design;
MONTECARLO:
NAMES ARE y x z;
BETWEEN ARE x z;
NOBSERVATIONS = 10000;
NCSIZES = 1;
CSIZES = 500(20);
SEED = 5724;
NREPS = 1;
SAVE IS sim.rcp.c.dat;
MODEL POPULATION:
%WITHIN%
yw BY y@1; yw*.7; y@.01;
%BETWEEN%
yb BY y@1; yb*.7; y@.01;
x*.7; z*.7; x WITH z*.1;
s | yb ON x*.1; s ON z*.1; s@.001;
yb ON z*.2; s WITH yb@0 x@0 z@0;
[yb*.1 s*.1];
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 5;
MODEL:
%WITHIN%
yw BY y@1; yw*.7; y@.01;
%BETWEEN%
yb BY y@1; yb*.7; y@.01;
x*.7; z*.7; x WITH z*.1;
s | yb ON x*.1; s ON z*.1; s@.001;
yb ON z*.2; s WITH yb@0 x@0 z@0;
[yb*.1 s*.1];
```

## Example (HS&B data)

### 15. *Syntax for B1 and B2 hypotheses using LMS*

```
TITLE: syntax for B1 and B2 hypotheses using LMS, 2x(1-1) design;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority ses mathach size sector;
USEVARIABLES ARE ses size mathach;
BETWEEN IS size; CLUSTER IS school;
DEFINE: size=size/1000;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 11;
MODEL:
%WITHIN%
sesw BY ses@1; sesw*.436; ses@.01;
mathachw BY mathach@1; mathachw*36.6; mathach@.01;
b1 | mathachw ON sesw;
sessizew | sesw XWITH size;
mathachw ON sessizew*.58;
%BETWEEN%
sesb BY ses@1; sesb*.15; ses@.01;
mathach*2.06; sessizeb | sesb XWITH size;
mathach ON size*-.11 sesb*7.1 sessizeb*-.52;
[ses@0 sesb*-.01 mathach*12.81 b1*1.6];
b1*.6; mathach WITH b1*-.23;
sesb WITH size*-.03; b1 WITH sesb*.07;
OUTPUT: TECH1 TECH3;
```

### 16. *Syntax for a B1 hypothesis using RCP and a B2 hypothesis using LMS*

```
TITLE: syntax for B1 using RCP and B2 using LMS, 2x(1-1) design;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority ses mathach size sector;
USEVARIABLES ARE ses size mathach;
BETWEEN IS size;
CLUSTER IS school;
DEFINE: size=size/1000;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 11;
MODEL:
%WITHIN%
sesw BY ses@1; sesw*.436; ses@.01;
s1 | mathach ON sesw; mathach*36.9;
%BETWEEN%
sesb BY ses@1; sesb*.15; ses@.01;
mathach*2.09; sessizeb | sesb XWITH size;
mathach ON size*-.1 sesb*7.12 sessizeb*-.585;
s1 ON size*.58; s1*.61; !cross-level interaction
[ses@0 sesb mathach*12.8 s1*1.6];
s1 WITH mathach*-.23;
sesb WITH size*-.03; s1 WITH sesb*.07;
```

17. *Syntax for conflated model*

```
TITLE: syntax for conflated model, 2x(1-1) design;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority ses mathach size sector;
USEVARIABLES ARE ses mathach size;
WITHIN IS ses; BETWEEN IS size; !"WITHIN IS" conflates effects
CLUSTER IS school;
DEFINE: size=size/1000;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
MODEL:
%WITHIN%
mathach*37; s1 | mathach ON ses;
%BETWEEN%
mathach*3.5; s1*; mathach WITH s1*;
mathach ON size*-.15; s1 ON size*2.22; [mathach*12.05 s1*2.39];
```

## Extensions (HS&B data)

18. *Syntax for a 3-way interaction of one L1 and two L2 variables (RCP and product)*

```
TITLE: syntax for 3-way interaction using RCP and product, 2x2x(1-1) design;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority ses mathach size sector;
USEVARIABLES ARE size ses sector mathach secsize;
BETWEEN IS size sector secsize;
CLUSTER IS school;
DEFINE: size=size/1000; secsize=sector*size;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS BAYES;
ALGORITHM IS GIBBS(RW);
MODEL:
%WITHIN%
sesw BY ses@1; sesw*.435; ses@.01;
mathach*36.6; s1 | mathach ON sesw;
%BETWEEN%
sesb BY ses@1; sesb*.15; ses@.01;
size*.39; sector*.24; secsize*.22;
size WITH sector*-.14 secsize*-.03 sesb*-.03;
sector WITH secsize*.19 sesb*.07; secsize WITH sesb*.057;
mathach*1.71; s1*.375; s1 WITH mathach*.21 sesb*;
mathach ON size*0 sector*-.02 sesb*5.8 secsize*1.46;
s1 ON size*.23 sector*-.85 secsize*1.34;
[ses@0 sesb*.2 size*1.09 sector*.43 secsize*.34 mathach*12.2 s1*2.55];
```

19. *Syntax for a 3-way interaction of one L2 and two L1 variables (LMS only)*

```
TITLE: syntax for 3-way interaction using LMS, 2x(1-1) design;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority ses mathach size sector;
USEVARIABLES ARE minority ses size mathach;
BETWEEN IS size;
CLUSTER IS school;
DEFINE: size=size/1000;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS MONTECARLO;
MODEL:
%WITHIN%
sesw BY ses@1; sesw*.45; ses@.01;
minw BY minority@1; minw*.19; minority@.01;
mathach*35.7;
sesminw | sesw XWITH minw;
sesminwsize | sesminw XWITH size;
sessize | sesw XWITH size;
minwsize | minw XWITH size;
mathach ON sesw*1.5 minw*-2.15 sesminw*-.87 minwsize*-.95 sessize*.49
sesminwsize*-.43;
%BETWEEN%
sesb BY ses@1; sesb*.12; ses@.01;
mathach*12.6; mathach ON size*.0 sesb*5.9;
[ses@0 sesb mathach*2.17];
```

20. *Syntax for an interaction of L1 and L2 components of the same variable*

```
TITLE: syntax for interaction of L1 and L2 components of the same variable;
DATA: FILE IS hsab.dat;
VARIABLE: NAMES ARE school minority ses mathach size sector;
USEVARIABLES ARE ses mathach;
CLUSTER IS school;
ANALYSIS: TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS MLR;
ALGORITHM IS INTEGRATION; INTEGRATION IS 8;
MODEL:
%WITHIN%
sesw BY ses@1; sesw*.4; ses@.01;
mathach*37;
s1 | mathach ON sesw;
%BETWEEN%
sesb BY ses@1; sesb*.3; ses@.01;
mathach*2; mathach ON sesb*;
s1 ON sesb*; s1*.74; s1 WITH mathach*-.3;
[ses@0 sesb*-.01 mathach*13 s1*2.2];
```